
  1 !***********************************************************************
  2 ! etmsus.mac
  3 !***********************************************************************
  4 ! macro to analyse natural frequencies of an ETM quad suspension 
  5 ! 
  6 !Dennis Coyne 28-nov-2005 
  7 ! 
  8 finish 
  9 /CLEAR,START 
 10 /COM,ANSYS MODEL OF A SINGLE ETM QUAD SUSPENSION CHAIN
 11 /PREP7
 12 /TITLE,ETM QUAD SUSPENSION
 13 !***********************************************************************
 14 !*     GEOMETRIC PARAMETERS
 15 !***********************************************************************
 16 ! values of parameters
 17 ! SI units (m,N)
 18 ! naming per T040072-01, plus extensions explained below
 19 ! coordinate system per T040072-02 with origin at center of top blades
 20 
 21 ! top blades (D040298)
 22 lnb=0.480
 23 anb=0.095
 24 bnb=0.010  ! blade tip width
 25 cnb=7.5*3.14159/180  ! blade centerline angle to y-axis (rad)
 26 hnb=0.00429
 27 nn0=0.25
 28 !*
 29 ! middle blades (D040297)
 30 l1b=0.4207
 31 a1b=0.059
 32 b1b=0.010  ! blade tip width
 33 c1b=7.5*3.14159/180  ! blade centerline angle to y-axis (rad)
 34 h1b=0.0046
 35 n0=0.200
 36 !*
 37 ! bottom blades (D040296)
 38 l2b=0.370
 39 a2b=0.049
 40 b2b=0.010  ! blade tip width
 41 c2b=7.5*3.14159/180  ! blade centerline angle to y-axis (rad)
 42 h2b=0.0042
 43 n2=0.140
 44 !*
 45 ! Top Wires
 46 ! zw1 is the height coordinate of the top wire attachment to the top mass
 47 ! zcg1 is the height coordinate of the c.g. of the top mass
 48 ! zb1 is the height coordinate of the neutral axis of the middle blades 
 49 nn1=0.090
 50 ln=0.450

 51 rn=0.00052
 52 dm=0.001
 53 dn=0.001
 54 zw1=-SQRT(ln**2-(nn0-nn1)**2)
 55 zcg1=zw1-dm
 56 zb1=zcg1-dn
 57 !*
 58 ! Middle Wires
 59 ! zw2 is the height coordinate of the middle wire attachment to the UI mass
 60 ! zcg2 is the height coordinate of the c.g. of the UI mass
 61 ! zb2 is the height coordinate of the neutral axis of the bottom blades 
 62 n1=0.060
 63 l1=0.3085
 64 r1=0.00035
 65 d0=0.001
 66 d1=0.001
 67 su=0.003
 68 zw2=zb1-SQRT(l1**2-(n0-n1)**2)
 69 zcg2=zw2-d0
 70 zb2=zcg2-d1
 71 !*
 72 ! Penultimate Wires
 73 ! zw3 is the height coordinate upper (penultimate) wire attachment to the PM
 74 ! zcg3 is the height coordinate of the c.g. of the PM
 75 ! zb3 is the height coordinate of lower (TM) wire attachment to the PM
 76 n3=0.1635
 77 l2=0.3400
 78 r2=0.00031
 79 d2=0.001
 80 d3=0.001
 81 si=0.003
 82 zw3=zb2-SQRT(l2**2-(n2-n3)**2)
 83 zcg3=zw3-d2
 84 zb3=zcg3-d3
 85 !*
 86 ! Test Mass Wires/Ribbons
 87 ! zw4 is the height coordinate TM wire attachment to the TM
 88 ! zcg4 is the height coordinate of the c.g. of the TM
 89 n4=0.1585
 90 n5=0.1585
 91 l3=0.6020
 92 r3=0.00022
 93 d4=0.001
 94 sl=0.015
 95 zw4=zb3-SQRT(l3**2-(n4-n5)**2)
 96 zcg4=zw4-d4
 97 
 98 !***********************************************************************
 99 !*     MASS PARAMETERS
100 !***********************************************************************

101 !* N.B.: The moments of inertia below (Inx, Iny, Inz) are values without
102 !        the blade springs included.
103 !  Top Mass
104 mn=22.1
105 Inx=0.0670
106 Iny=0.4289
107 Inz=0.4309
108 !*
109 !  UI Mass
110 m1=21.8
111 I1x=0.0581
112 I1y=0.5005
113 I1z=0.4988
114 !*
115 !  PM
116 m2=38.4
117 I2x=0.2765
118 I2y=0.2752
119 I2z=0.4704
120 !*
121 !  TM
122 m3=39.61
123 I3x=0.2692
124 I3y=0.2679
125 I3z=0.4556
126 
127 !***********************************************************************
128 !*     MATERIAL PROPERTIES
129 !***********************************************************************
130 maryoung=1.76e11 
131 marpoiss=0.3 
132 mardens=7800 
133 wireyoung=2.0e11 
134 wirepoiss=0.3 
135 wiredens=7800 
136 
137 !MPTEMP,,,,,,,, 
138 !MPTEMP,1,0 
139 !MPDATA,EX,1,,maryoung 
140 !MPDATA,PRXY,1,,marpoiss 
141 !MPDATA,DENS,1,,mardens 
142 !MPTEMP,,,,,,,, 
143 !MPTEMP,1,0 
144 !MPDATA,EX,2,,wireyoung 
145 !MPDATA,PRXY,2,,wirepoiss 
146 !MPTEMP,,,,,,,, 
147 !MPTEMP,1,0 
148 !MPDATA,DENS,2,,wiredens 
149 MP,EX,1,maryoung
150 MP,EY,1,maryoung

151 MP,EZ,1,maryoung
152 MP,PRXY,1,marpoiss
153 MP,DENS,1,mardens
154 MP,EX,2,wireyoung
155 MP,EY,2,wireyoung
156 MP,EZ,2,wireyoung
157 MP,PRXY,2,wirepoiss
158 MP,DENS,2,wiredens
159 
160 !***********************************************************************
161 !*     GENERATE GEOMETRY
162 !***********************************************************************
163 !*          BLADES
164 !***********************************
165 ! Top Blade, Right
166 K,1,0,nn0,0, 
167 K,2,-bnb*cos(cnb)/2,nn0-bnb*sin(cnb)/2,0, 
168 K,3,bnb*cos(cnb)/2,nn0+bnb*sin(cnb)/2,0, 
169 K,4,lnb*sin(cnb),nn0-lnb*cos(cnb),0, 
170 K,5,lnb*sin(cnb)-anb*cos(cnb)/2,nn0-lnb*cos(cnb)-anb*sin(cnb)/2,0, 
171 K,6,lnb*sin(cnb)+anb*cos(cnb)/2,nn0-lnb*cos(cnb)+anb*sin(cnb)/2,0, 
172 LSTR, 1, 4 ! line 1
173 LSTR, 4, 6 ! line 2
174 LSTR, 6, 3 ! line 3
175 LSTR, 3, 1 ! line 4
176 LSTR, 4, 5 ! line 5
177 LSTR, 5, 2 ! line 6
178 LSTR, 2, 1 ! line 7
179 AL,1,2,3,4 ! area 1
180 AL,1,5,6,7 ! area 2
181 !*
182 ! Top Blade, Left
183 K,7,0,-nn0,0, 
184 K,8,bnb*cos(cnb)/2,-nn0+bnb*sin(cnb)/2,0, 
185 K,9,-bnb*cos(cnb)/2,-nn0-bnb*sin(cnb)/2,0, 
186 K,10,-lnb*sin(cnb),-nn0+lnb*cos(cnb),0, 
187 K,11,-lnb*sin(cnb)+anb*cos(cnb)/2,-nn0+lnb*cos(cnb)+anb*sin(cnb)/2,0, 
188 K,12,-lnb*sin(cnb)-anb*cos(cnb)/2,-nn0+lnb*cos(cnb)-anb*sin(cnb)/2,0, 
189 LSTR, 7,10 ! line 8
190 LSTR,10,12 ! line 9
191 LSTR, 9,12 ! line 10
192 LSTR, 7, 9 ! line 11
193 LSTR,10,11 ! line 12
194 LSTR, 8,11 ! line 13
195 LSTR, 7, 8 ! line 14
196 AL,8,9,10,11 ! area 3
197 AL,8,12,13,14 ! area 4
198 !*
199 ! Middle Blade, Right
200 K,13,0,n0,zb1, 

201 K,14,-b1b*cos(c1b)/2,n0-b1b*sin(c1b)/2,zb1, 
202 K,15,b1b*cos(c1b)/2,n0+b1b*sin(c1b)/2,zb1, 
203 K,16,l1b*sin(c1b),n0-l1b*cos(c1b),zb1, 
204 K,17,l1b*sin(c1b)-a1b*cos(c1b)/2,n0-l1b*cos(c1b)-a1b*sin(c1b)/2,zb1, 
205 K,18,l1b*sin(c1b)+a1b*cos(c1b)/2,n0-l1b*cos(c1b)+a1b*sin(c1b)/2,zb1, 
206 LSTR, 13,16 ! line 15
207 LSTR, 16,18 ! line 16
208 LSTR, 18,15 ! line 17
209 LSTR, 15,13 ! line 18
210 LSTR, 16,17 ! line 19
211 LSTR, 17,14 ! line 20
212 LSTR, 14,13 ! line 21
213 AL,15,16,17,18 ! area 5
214 AL,15,19,20,21 ! area 6
215 !*
216 ! Middle Blade, Left
217 K,19,0,-n0,zb1, 
218 K,20,b1b*cos(c1b)/2,-n0+b1b*sin(c1b)/2,zb1, 
219 K,21,-b1b*cos(c1b)/2,-n0-b1b*sin(c1b)/2,zb1, 
220 K,22,-l1b*sin(c1b),-n0+l1b*cos(c1b),zb1, 
221 K,23,-l1b*sin(c1b)+a1b*cos(c1b)/2,-n0+l1b*cos(c1b)+a1b*sin(c1b)/2,zb1, 
222 K,24,-l1b*sin(c1b)-a1b*cos(c1b)/2,-n0+l1b*cos(c1b)-a1b*sin(c1b)/2,zb1, 
223 LSTR, 19,22 ! line 22
224 LSTR, 22,24 ! line 23
225 LSTR, 21,24 ! line 24
226 LSTR, 19,21 ! line 25
227 LSTR, 22,23 ! line 26
228 LSTR, 20,23 ! line 27
229 LSTR, 19,20 ! line 28
230 AL,22,23,24,25 ! area 7
231 AL,22,26,27,28 ! area 8
232 !*
233 ! Bottom Blade, Right
234 K,25,0,n2,zb2, 
235 K,26,-b2b*cos(c2b)/2,n2-b2b*sin(c2b)/2,zb2, 
236 K,27,b2b*cos(c2b)/2,n2+b2b*sin(c2b)/2,zb2, 
237 K,28,l2b*sin(c2b),n2-l2b*cos(c2b),zb2, 
238 K,29,l2b*sin(c2b)-a2b*cos(c2b)/2,n2-l2b*cos(c2b)-a2b*sin(c2b)/2,zb2, 
239 K,30,l2b*sin(c2b)+a2b*cos(c2b)/2,n2-l2b*cos(c2b)+a2b*sin(c2b)/2,zb2, 
240 LSTR, 25,28 ! line 29
241 LSTR, 28,30 ! line 30
242 LSTR, 30,27 ! line 31
243 LSTR, 27,25 ! line 32
244 LSTR, 28,29 ! line 33
245 LSTR, 29,26 ! line 34
246 LSTR, 26,25 ! line 35
247 AL,29,30,31,32 ! area 9
248 AL,29,33,34,35 ! area 10
249 !*
250 ! Bottom Blade, Left

251 K,31,0,-n2,zb2, 
252 K,32,b2b*cos(c2b)/2,-n2+b2b*sin(c2b)/2,zb2, 
253 K,33,-b2b*cos(c2b)/2,-n2-b2b*sin(c2b)/2,zb2, 
254 K,34,-l2b*sin(c2b),-n2+l2b*cos(c2b),zb2, 
255 K,35,-l2b*sin(c2b)+a2b*cos(c2b)/2,-n2+l2b*cos(c2b)+a2b*sin(c2b)/2,zb2, 
256 K,36,-l2b*sin(c2b)-a2b*cos(c2b)/2,-n2+l2b*cos(c2b)-a2b*sin(c2b)/2,zb2, 
257 LSTR, 31,34 ! line 36
258 LSTR, 34,36 ! line 37
259 LSTR, 33,36 ! line 38
260 LSTR, 31,33 ! line 39
261 LSTR, 34,35 ! line 40
262 LSTR, 32,35 ! line 41
263 LSTR, 31,32 ! line 42
264 AL,36,37,38,39 ! area 11
265 AL,36,40,41,42 ! area 12
266 !***********************************
267 !*          MASSES
268 !***********************************
269 !*
270 ! Top Mass
271 K,100,0,0,zcg1,
272 K,101,0,nn1,zw1,
273 K,102,0,-nn1,zw1,
274 K,103,su,n0,zb1,
275 K,104,-su,n0,zb1,
276 K,105,su,-n0,zb1,
277 K,106,-su,-n0,zb1,
278 
279 !*
280 ! Upper Intermediate (UI) Mass
281 K,200,0,0,zcg2,
282 K,201,su,n1,zw2,
283 K,202,-su,n1,zw2,
284 K,203,su,-n1,zw2,
285 K,204,-su,-n1,zw2,
286 K,205,si,n2,zb2,
287 K,206,-si,n2,zb2,
288 K,207,si,-n2,zb2,
289 K,208,-si,-n2,zb2,
290 
291 !*
292 ! Penultimate Mass (PM)
293 K,300,0,0,zcg3,
294 K,301,si,n3,zw3,
295 K,302,-si,n3,zw3,
296 K,303,si,-n3,zw3,
297 K,304,-si,-n3,zw3,
298 K,305,sl,n4,zb3,
299 K,306,-sl,n4,zb3,
300 K,307,sl,-n4,zb3,

301 K,308,-sl,-n4,zb3,
302 
303 !*
304 ! Test Mass (TM)
305 K,400,0,0,zcg4,
306 K,401,sl,n5,zw4,
307 K,402,-sl,n5,zw4,
308 K,403,sl,-n5,zw4,
309 K,404,-sl,-n5,zw4,
310 
311 !***********************************
312 !*          WIRES
313 !***********************************
314 !*
315 ! Top Wires
316 LSTR,1,101     ! line 43
317 LSTR,7,102     ! line 44
318 !*
319 ! UI Wires
320 LSTR,103,201   ! line 45
321 LSTR,104,202   ! line 46
322 LSTR,105,203   ! line 47
323 LSTR,106,204   ! line 48
324 !*
325 ! PM Wires
326 LSTR,205,301   ! line 49
327 LSTR,206,302   ! line 50
328 LSTR,207,303   ! line 51
329 LSTR,208,304   ! line 52
330 !*
331 ! TM Wires
332 LSTR,305,401   ! line 53
333 LSTR,306,402   ! line 54
334 LSTR,307,403   ! line 55
335 LSTR,308,404   ! line 56
336 
337 !***********************************
338 !*          RIGID LINKS
339 !***********************************
340 !*
341 ! Top Mass
342 LSTR,100,101   ! line 57, top wire 1
343 LSTR,100,102   ! LINE 58, top wire 2
344 LSTR, 15,103   ! LINE 59, middle blade to wire 1
345 LSTR, 14,104   ! LINE 60, middle blade to wire 2
346 LSTR, 20,105   ! LINE 61, middle blade to wire 3
347 LSTR, 21,106   ! LINE 62, middle blade to wire 4
348 LSTR,100, 16   ! LINE 63, c.g. to blade root 1
349 LSTR,100, 17   ! LINE 64, c.g. to blade root 2
350 LSTR,100, 18   ! LINE 65, c.g. to blade root 3

351 LSTR,100, 22   ! LINE 66, c.g. to blade root 4
352 LSTR,100, 23   ! LINE 67, c.g. to blade root 5
353 LSTR,100, 24   ! LINE 68, c.g. to blade root 6
354 !*
355 ! UI Mass
356 LSTR,200,201   ! line 69, UI wire 1
357 LSTR,200,202   ! LINE 70, UI wire 2
358 LSTR,200,203   ! LINE 71, UI wire 3
359 LSTR,200,204   ! LINE 72, UI wire 4
360 LSTR, 27,205   ! LINE 73, bottom blade to wire 1
361 LSTR, 26,206   ! LINE 74, bottom blade to wire 2
362 LSTR, 32,207   ! LINE 75, bottom blade to wire 3
363 LSTR, 33,208   ! LINE 76, bottom blade to wire 4
364 LSTR,200, 28   ! LINE 77, c.g. to blade root 1
365 LSTR,200, 29   ! LINE 78, c.g. to blade root 2
366 LSTR,200, 30   ! LINE 79, c.g. to blade root 3
367 LSTR,200, 34   ! LINE 80, c.g. to blade root 4
368 LSTR,200, 35   ! LINE 81, c.g. to blade root 5
369 LSTR,200, 36   ! LINE 82, c.g. to blade root 6
370 !*
371 ! PM
372 LSTR,300,301   ! LINE 83, c.g. to PM wire 1
373 LSTR,300,302   ! LINE 84, c.g. to PM wire 2
374 LSTR,300,303   ! LINE 85, c.g. to PM wire 3
375 LSTR,300,304   ! LINE 86, c.g. to PM wire 4
376 LSTR,300,305   ! LINE 87, c.g. to TM wire 1
377 LSTR,300,306   ! LINE 88, c.g. to TM wire 2
378 LSTR,300,307   ! LINE 89, c.g. to TM wire 3
379 LSTR,300,308   ! LINE 90, c.g. to TM wire 4
380 !*
381 ! TM
382 LSTR,400,401   ! LINE 91, c.g. to TM wire 1
383 LSTR,400,402   ! LINE 92, c.g. to TM wire 2
384 LSTR,400,403   ! LINE 93, c.g. to TM wire 3
385 LSTR,400,404   ! LINE 94, c.g. to TM wire 4
386 !*
387 ! Plot Areas
388 aplot
389 
390 !***********************************************************************
391 !*     MESH GEOMETRY
392 !***********************************************************************
393 !*
394 !***********************************
395 !           Blade Springs
396 !***********************************
397 !ET,1,SHELL63,0,0,1,,0,0
398 !ET,1,SHELL43,,,2
399 !ET,1,SHELL181,1
400 !ET,1,SHELL93

401 ET,1,SHELL63
402 TYPE,1     ! shell elements
403 MAT,1      ! maraging steel
404 !     Top Blades
405 R,1,hnb               !for the blade 
406 REAL,1
407 ESIZE,anb/4,0 
408 amesh,1,4
409 !     Middle Blades
410 R,2,h1b               !for the blade 
411 REAL,2
412 ESIZE,a1b/4,0
413 amesh,5,8
414 !     Bottom Blades
415 R,3,h2b               !for the blade 
416 REAL,3
417 ESIZE,a2b/4,0
418 amesh,9,12
419 
420 !***********************************
421 !           Suspension Masses
422 !***********************************
423 ET,2,MASS21,0,0,0
424 TYPE,2     ! discrete mass elements
425 !     Top Mass
426 R,4,mn,mn,mn,Inx,Iny,Inz
427 REAL,4  
428 KMESH,100
429 !     UI Mass
430 R,5,m1,m1,m1,I1x,I1y,I1z
431 REAL,5  
432 KMESH,200
433 !     PM
434 R,6,m2,m2,m2,I2x,I2y,I2z
435 REAL,6  
436 KMESH,300
437 !     TM
438 R,7,m3,m3,m3,I3x,I3y,I3z
439 REAL,7  
440 KMESH,400
441 
442 ! Wire Clamps
443 !R,8,mc,mc,mc,Icx,Icy,Icz
444 !REAL,8
445 !kmesh,?
446 
447 !***********************************
448 !           Wires
449 !***********************************
450 ET,3,BEAM4,,0,,,,0

451 TYPE,3     ! beam elements
452 MAT,2
453 !     Top Wires
454 wAn=3.14159*rn**2
455 wIn=(3.14159*rn**4)/4
456 wJn=(3.14159*rn**4)/2
457 R,9,wAn,wIn,wIn,rn,rn,0
458 RMORE,0,wJn
459 REAL,9
460 ESIZE,0.005
461 LMESH,43,44
462 !     UI Wires
463 wA1=3.14159*r1**2
464 wI1=(3.14159*r1**4)/4
465 wJ1=(3.14159*r1**4)/2
466 R,10,wA1,wI1,wI1,r1,r1,0
467 RMORE,0,wJ1
468 REAL,10
469 ESIZE,0.005
470 LMESH,45,48
471 !     PM Wires
472 wA2=3.14159*r2**2
473 wI2=(3.14159*r2**4)/4
474 wJ2=(3.14159*r2**4)/2
475 R,11,wA2,wI2,wI2,r2,r2,0
476 RMORE,0,wJ2
477 REAL,11
478 ESIZE,0.005
479 LMESH,49,52
480 !     TM Wires
481 wA3=3.14159*r3**2
482 wI3=(3.14159*r3**4)/4
483 wJ3=(3.14159*r3**4)/2
484 R,12,wA3,wI3,wI3,r3,r3,0
485 RMORE,0,wJ3
486 REAL,12
487 ESIZE,0.005
488 LMESH,53,56
489 
490 !***********************************
491 !           Rigid Links
492 !***********************************
493 ET,4,MPC184,1,0
494 TYPE,4      ! rigid body constraint elements
495 ESIZE,,1
496 LMESH,57,94
497 
498 NUMMRG,NODE,0.00001
499 
500 !***********************************************************************

501 !*     BOUNDARY CONDITIONS
502 !***********************************************************************
503 !*
504 ! Clamp the upper blade roots
505 DL, 2, ,ALL,0  
506 DL, 5, ,ALL,0  
507 DL, 9, ,ALL,0
508 DL,12, ,ALL,0
509 EPLOT
510 
511 FINISH
512 
513 !***********************************************************************
514 !   Static Preload Analysis
515 !***********************************************************************
516 /SOL
517 ANTYPE,STATIC
518 PSTRES,ON
519 G=9.84
520 ACEL,0,0,G
521 SOLVE
522 FINISH
523 !/POST1
524 !PLDIST,0
525 !FINISH
526 !***********************************************************************
527 !   Eigenvalue Analysis
528 !***********************************************************************
529 /SOL
530 ANTYPE,MODAL
531 MSAVE,0 
532 !*  
533 MODOPT,LANB,60  
534 EQSLV,JCG 
535 !MXPAND,0, , ,0  
536 LUMPM,OFF
537 PSTRES,ON
538 !*  
539 /STATUS,SOLU
540 SOLVE   
541 FINISH  
542 /POST1  
543 /output,etmsus,out,,
544 !*Status,argx
545 SET,LIST
546 /output
547 *get,freq1,mode,1,freq
548 *get,freq2,mode,2,freq
549 *get,freq3,mode,3,freq
 


